
1 Overview 

As  of  ECO  IV  it  is  possible  to  create  VCL.NET  applications  that  utilize  the  ECO  framework.  Instances  of  ECO  modeled
classes may be manipulated via various TDataSet descendant components. The main components are as follows:

TReferenceHandle

Holds a reference to an EcoSpace which other handles depend upon for fetching data etc. This component may also hold a
reference to an instance of a modeled class, a basic type such as an Integer or String, or a collection.

TExpressionHandle

Obtains an EcoSpace reference via its RootHandle property. The "Expression" property is evaluated in memory in order to
determine what type of value the handle holds (a collection of objects, etc).

TOclPsHandle

Obtains  an  EcoSpace  reference  via  its  RootHandle  property.  The  "Expression"  property  is  evaluated  in  the  persistence
storage rather than in memory, in the case of an RDBMS persistence storage the expression will be evaluated as SQL.

2 Prerequisites and goals 

Prerequisites

To successfully follow this document the user should have read the preceding articles in this series and have a copy of the
model created.

Goals

By the end of this document you will be able to

• Create ECO VCL.NET applications

• Link standard VCL.NET DB controls to instances of ECO modeled classes.

• Work with multiple EcoSpaces to mimic database transactions.

3 Creating the application 

1. From the File->New menu select Other.

2. In the TreeView on the left hand side select the "ECO" node.

3. Select the "ECO Delphi.NET Project Wizard" and click OK.

4. In the Presets combobox select VCL.NET application.

5. For the Project Name enter QuickStartVCL.

6. Specify a location for your new project.

7. In the "Model" group deselect "Add a model with empty package". This option is used to create the model within the same 
project. For this sample we will be using the model created in earlier articles. In addition it is good practise to have a 
separate project for business classes so that more than one GUI project may use the same classes, as we will see in a 
later article when ASP.NET is covered.

8. In the "Persistence" group ensure that the "Storage" is set to XML. XML is useful for getting started quickly, a proper 
database will be used in a later article where various database features will be explained.

9. Finally ensure that "Use a PersistenceMapperProvider" is selected, and then click OK.

4 Understanding the project files 

The ECO VCL.NET application has now been created from a template and contains the following files:

EcoSpaceUnit.pas

This unit contains a component similar to a DataModule named "QuickStartVCLEcoSpace". This component will be used to
describe the complete business model by specifying one or more ECO packages at design time. Packages can be thought
of  as  application  building  blocks  consisting  of  one  or  more  classes.  ECO  packages  may  be  used  in  more  than  one
application.

It  is also possible to drop components onto this EcoSpace. This is how persistence is specified. It  is quite common for the
persistence  components  to  be  added  directly  to  the  EcoSpace.  Because  this  article  will  demonstrate  the  technique  of
creating  multiple  EcoSpace  instances  a  way  is  needed  to  share  the  persistence  components  (and  therefore  share  the
database  connection).  This  is  the  purpose  of  the  component  "PersistenceMapperSharer1"  that  is  visible  on  the  design
surface  of  the  EcoSpace,  and  is  the  result  of  selecting  the  "Use  a  PersistenceMapperProvider"  checkbox  on  the  ECO
Delphi.NET Project Wizard.

The  PersistenceMapperSharer1  has  a  property  named  "MapperProviderTypeName",  this  identifies  the  class  within  the
PersistenceMapperProvider.pas file described next.

PersistenceMapperProvider.pas

This  unit  contains  the  class  that  owns  the  database  connection  component  and  ECO  persistence  components.  A  single
instance of this class will be created automatically by ECO at runtime and all persistence requests will be served by this one
instance.  Only  a  single  connection  to  the  database  is  created,  regardless  of  how  many  instances  of  the  EcoSpace  are
created. Persistence requests are also thread safe.

Due  to  the  settings  specified  in  the  project  wizard  you  should  see  only  a  single  component  on  the  design  surface  of  this
component, "persistenceMapperXml1".

Unit1

This unit contains the main form of the application. The form has some ECO specific code in it, which is as follows:

1. The form has a component named rhRoot on it. This component will be used to hold a reference to an instance of the 
application's EcoSpace. In addition its EcoSpaceTypeName identifies the class name of the EcoSpace. This is used at 
design time by the ECO component and property editors. Once the project is compiled these editors can inspect its binary 
to find model information. This is possible because the ECO packages used by the EcoSpace are specified using .NET 
attributes and are therefore available via reflection.

2. EcoSpace property: It is a programming shortcut so that "EcoSpace" can be typed instead of "rhRoot.EcoSpace". In 
addition the property will be of the type QuickStartVclEcoSpace rather than the DefaultEcoSpace type exposed via 
rhRoot.EcoSpace.

3. There is also an overloaded constructor which takes an additional EcoSpace parameter. Instead of creating a new 
instance of the EcoSpace type (which is what the standard constructor does) this overloaded constructor will use the 
EcoSpace instance passed to it.

NOTE: It  is  possible to have more than one type of  EcoSpace in an application.  Each type using different ECO Packages
and updating different persistence stores.

5 Specifying the business classes 

The next step is to identify which ECO Packages the application's EcoSpace should use.

1. In the project manager right-click the project group and select "Add existing project".

2. Add the CapableObjects.QuickStart.BusinessModel project.

3. Right-click the newly added project and select "Build sooner" from the context menu, to ensure that the business classes 
are always compiled before this application.

4. In the project manager right-click the "References" node belonging to the QuickStartVCL project and select "Add 
reference".

5. Select the [Project references] tab.

6. Click "Add reference".

7. Select CapableObject.QuickStart.BusinessModel.

8. Click OK.

Because a single project may consist of many types of EcoSpace using different ECO packages it is not enough to merely
add the classes to the References of the project. To illustrate this point select the rhRoot component on Form1 and click the
[...]  editor  button  on  its  property  "StaticValueTypeName".  On the  form that  appears  expand the  nodes as  illustrated  in  the
screenshot below, this will reveal there are no modeled classes available.

To associate the modeled package with this EcoSpace follow these steps:

1. Compile all projects. This ensures that the binaries are available for the ECO designers to read for model information.

2. Click the "Select Packages" button at the bottom of the design surface for the EcoSpace. (Double-click the EcoSpaceUnit, 
and then look for the "Design" tab at the bottom of the editor).

3. In the dialog that appears drag the package named in the left panel (Available UML Packages) and drop it into the panel 
on the right (Selected UML Packages).

4. Now build the project again. This is necessary because the binary previously built does not reflect the changes just made.

Viewing the editor for the StaticValueTypeName property now shows the following.

NOTE: If you still do not see any classes try closing the editor tab for the form, recompiling, and then reopening.

6 How models are constructed 

When design time support is needed for an ECO component, or runtime model information is required in order to perform a
task, the model is obtained from the EcoSpace.

A single application may contain more than one type of EcoSpace. To define the model for an EcoSpace the developer must
select one or more ECO Packages. ECO Packages may be used in more than one EcoSpace type, and in more than one
application.  In  the  diagram  above  both  EcoSpace  types  use  the  CommonPackage,  in  addition  to  this  both  the
AccountsPackage and the PersonnelPackage have classes that descend from classes within the CommonPackage.

When Form1 requires model information the ECO designers will discover the following classes: Class1, Class2, BaseClass1,
BaseClass2.  When Form2 requires  model  information  the  ECO designers  will  discover  the  following classes:  BaseClass1,
BaseClass2, Class3, Class4.

7 Creating and modifying objects 

1. Drop a TExpressionHandle onto the form.

2. Name it "ehAuthors".

3. Set its RootHandle property to point to rhRoot. This property has two purposes:

1. At design time the TExpressionHandle can find the EcoSpaceTypeName and therefore obtain model information.

2. At runtime the TExpressionHandle can obtain a reference to the EcoSpace instance in order to create instances of 
modeled business classes, delete instances, or modify their property (column) values.

4. Click the [...] editor on the expression handle's "Expression" property.

5. In the editor enter the expression

Author.allInstances

6. The OCL editor will confirm that the expression is valid, and at the bottom indicate that the result of the expression is 
"Collection(Author)".

7. Now that ehAuthors has a context (a collection of Author instances) right-click the component and select "Create default 
columns".

8. In the Object Inspector set AddDefaultProperties to False. At runtime a Column will be added for each property on the 
class, as we want full control over which columns are created this should be set to False.

9. Click the [...] editor on the Columns property and remove Articles, FullName, and ID.

TReferenceHandle,  TExpressionHandle,  TOclPsHandle,  and  TVariableHandle  all  ultimately  descend  from  TDataSet.  This
allows us to use standard DB components and controls to manipulate the object instances within the EcoSpace.

10. Add a DataSource to the form.

11. Name it "dsAuthors".

12. Set its DataSet property to point to ehAuthors.

13. Add a TDBGrid to the form.

14. Name it "grdAuthors".

15. Set its DataSource property to dsAuthors.

16. Add a TDBNavigator to the form.

17. Name it "navAuthors".

18. Set its DataSource property to dsAuthors.

8 Running the application for the first 
time 

1. Run the application.

2. Enter one or two rows of data.

3. Close the application.

4. Now restart the application and you will notice that the data you previously entered has now disappeared.

This  is  because  when  you  Post  a  row  you  are  merely  committing  it  to  the  EcoSpace,  no  data  base  updates  have  been
performed.  This  is  similar  to  the  way  a  DB  transaction  would  work  where  you  must  explicitly  commit  the  changes  the
occurred within it.

5. Add a button to the form.

6. Name it "btnCommit".

7. Set its caption to "Commit".

8. Set its OnClick event to the following code:

procedure TForm1.btnCommitClick(Sender: TObject);
begin
  EcoSpace.UpdateDatabase;
end;

Now  run  the  application  again,  enter  two  rows  of  data,  but  before  closing  the  form  click  the  Commit  button.  Close  the
application and then rerun it again, this time the data has survived the life of the application.

NOTE: The data has been persisted into a file named Data.xml in the application's folder.

9 Creating a secondary form for editing 
articles 

Next  a  secondary  form will  be  created.  This  form will  accept  an  instance  of  the  Author  class  and  display  their  articles  for
editing.

1. Select the File->New->Other menu.

2. In the TreeView on the left hand side select the "ECO" node.

3. Select "New ECO VCL.NET form" and click OK.

4. Name the new form "fmAuthorArticles" and save it as AuthorArticlesForm.

5. Remove the form from the project's list of forms to auto-create when the application starts (Project->Options menu).

6. Add QuickStartPackageUnit and EcoSpaceUnit to the uses clause of the interface section.

7. On the rhRoot component select QuickStartVCLEcoSpace in the drop down list for the EcoSpaceTypeName property.

8. Add a new constructor like so:

constructor Create(AOwner: TComponent; AEcoSpace: QuickStartVCLEcoSpace; SelectedAuthor: 
Author); overload;
 
...
 
constructor TfmAuthorArticles.Create(AOwner: TComponent; AEcoSpace: 
QuickStartVCLEcoSpace; SelectedAuthor: Author);
begin
  inherited Create(AOwner);
  SetEcoSpace(AEcoSpace);
  rhRoot.SetElement(SelectedAuthor);
end;

and change the type of the EcoSpace property to QuickStartVCLEcoSpace

private
  FEcoSpace: QuickStartVCLEcoSpace;
  ...
public
  ...
  property EcoSpace: QuickStartVCLEcoSpace read FEcoSpace;
end;

9. Switch back to Unit1.

10. Add AuthorArticlesForm and QuickStartPackageUnit to the uses clause of the implementation section.

11. Add a button to the form.

12. Name it btnShowArticles and set its caption "Show articles".

13. In its OnClick event add the following code:

procedure TForm1.btnShowArticlesClick(Sender: TObject);
var
  SelectedAuthor: Author;
  AuthorArticlesForm: TfmAuthorArticles;
begin
  //Make sure the row is saved first
  if ehAuthors.State <> dsBrowse then
  begin
    ShowMessage('Please save or cancel your changes first');
    Exit;
  end;
 
  //If there are no rows then do not continue
  if ehAuthors.CurrentElement = nil then
    Exit;
 
  //Use AsObject to get an object that can be cast to Author
  SelectedAuthor := ehAuthors.CurrentElement.AsObject as Author;
 
  //Create the form and show it modally
  AuthorArticlesForm := TfmAuthorArticles.Create(Self, EcoSpace, SelectedAuthor);
  try
    AuthorArticlesForm.ShowModal;
  finally
    AuthorArticlesForm.Free;
  end;
end;

Next some GUI will be added to the articles form.

14. Switch back to the AuthorArticlesForm.

15. Select the rhRoot component.

16. In the StaticValueTypeName enter Author (or select it using the property editor). The purpose of this is twofold

1. It helps the ECO designers at design time by informing it what type of object the component will hold, this is used for 
example whenever the OCL editor is invoked.

2. At runtime it is used to ensure that any reference the component holds (via SetElement) is compatible with the type 
specified, ensuring that all OCL expressions entered at design time will continue to work.

17. Add a TExpressionHandle to the form.

18. Name it "ehAuthorArticles".

19. Set its RootHandle property to point to rhRoot.

20. Set its expression to self.Articles - this will confirm the expression is valid and indicate the return type will be a collection 
of Article.

21. Set AddDefaultProperties to False.

22. Right-click the component and select "Create default columns".

23. Click the [...] property editor on the Columns property, remove the ID and Author columns.

24. Ensure that the column for ReasonRejected is set to be read-only.

25. Add a DataSource to the form.

26. Name it dsAuthorArticles.

27. Set its DataSet property to ehAuthorArticles.

28. Now add a TDBGrid.

29. Name it grdAuthorArticles.

30. Set its DataSource to ehAuthorArticles.

31. You may optionally edit the columns of grdAuthorArticles to show only Title, MainState, ReviewState, and 
ReasonRejected.

32. Add a TDBNavigator to the form.

33. Set its name to navAuthorArticles.

34. Set its DataSource to point to dsAuthorArticles.

Now run the application. Select the first of the two authors you entered earlier and click the "Show articles" button and add
an article or two for this author.

Because this form uses the same EcoSpace as the main form it is necessary to click the Commit button on the main form if
you wish to commit your changes to the data store.

10 Adding more controls to the articles 
form 

Next some more controls will be added to the articles form to allow the user to enter more information about the article.

1. First expand the Options property of the TDBGrid and ensure that dgCancelOnExit is set to False. If this is not done then 
it is standard VCL behavior to remove any newly inserted row when the grid loses focus.

2. Add a TDBEdit to the form.

3. Name it "edTitle".

4. Link the DataSource property to dsAuthorArticles.

5. Link the DataField property to the Title column.

6. Add a TDBMemo named memDescription and link it to the Description column.

7. Add another TDBMemo named memBody and link it to the Body column.

11 Observing state machine behavior 

Next the user will be able to submit their article for publication. We will observe how the object behaves differently depending
on the value of Author.AutoPublishArticles.

1. Drop a Button onto the form.

2. Name it btnPublish.

3. In the OnClick event enter the following source code:

procedure TfmArticles.btnPublishClick(Sender: TObject);
var
  SelectedArticle: Article;
begin
  //If no article is present then exit
  if ehAuthorArticles.CurrentElement = nil then
    Exit;
 
  //Only allow if the article is not being edited
  if ehAuthorArticles.State <> dsBrowse then
  begin
    ShowMessage('Please save or cancel your changes first');
    Exit;
  end;
 
  //Use AsObject to get an object that can be cast to an Article
  SelectedArticle := ehAuthorArticles.CurrentElement.AsObject as Article;
  try
    //Invoke the Publish trigger
    SelectedArticle.Publish;
  except
    //If there is no transition out of the current state that is
    //trigger by the Publish trigger then trap the exception
    //and show a friendly message
    on E: NoTransitionException do
      ShowMessage('Cannot publish an article in this state');
  end;
end;

4. Add Eco.ObjectRepresentation to the uses clause in the implementation section, this is where NoTransitionException is 
defined.

5. Run the application again.

6. Ensure that there is one Author with AutoPublishArticles set to True. Select this author and click "Show articles".

7. Select an existing article, or create a new article and save it.

8. Click the Publish button. You will now observe the MainState has changed to Published.

9. Close the articles form.

10. Select the Author with AutoPublishArticles set to False and click "Show articles".

11. Again ensure you have an article for this author and click the Publish button.

12. This time you will see that the MainState has changed to Reviewing and the ReviewState has changed to 
CheckingSpelling.

12 Creating a custom column 

Next a custom column will be added to the Author class indicating how many articles the author has written.

1. Select the ehAuthors component on the main form.

2. Click the [...] property editor on the Columns property.

3. Add a new column, name it "NumberOfArticles".

4. Set IsReadOnly to True.

5. Click the [...] property editor for the Expression property.

6. In the OCL editor enter the following expression:

self.Articles->size

7. Now run the application again.

8. Make sure that the NumberOfArticles column is in view in the DBGrid.

9. Select an author and click the "Show articles" button.

10. Add a new article and click the Post button on the navigator.

11. You will observe NumberOfArticles updates automatically.

13 Using multiple EcoSpace instances 

There  are  various  ways  of  isolating  changes  made  within  forms  from  other  forms  in  the  application,  allowing  the  user  to
commit or revert their changes without affecting the data in other forms. One of these techniques is to use a new instance of
the application's EcoSpace class in each form. This technique will be demonstrated next.

1. Change the constructor of fmAuthorArticles to the following:

constructor Create(AOwner: TComponent; SelectedAuthor: Author); overload;
 
...
 
constructor TfmArticles.Create(AOwner: TComponent; SelectedAuthor: Author);
begin
  //Call the base constructor
  inherited Create(AOwner);
 
  //Create a new EcoSpace for this form to use and activate it
  SetEcoSpace(QuickStartVCLEcoSpace.Create);
  EcoSpace.Active := True;
 
  //Set the element to the author that was passed from the main form
  rhRoot.SetElement(SelectedAuthor);
end;

2. Now change the source code in the main form that creates an instance of this form:

AuthorArticlesForm := TfmAuthorArticles.Create(Application, SelectedAuthor);

3. Now run the application.

4. Select an author and click "Show articles".

At this point you should experience an exception informing you that the value you are trying to use in rhRoot.SetElement is
an object that belongs to another EcoSpace. The reason you cannot mix EcoSpaces is to avoid problems with associations.
If ObjectA belonged to instance A of an EcoSpace and ObjectB belonged to instance B of an EcoSpace, ECO would prevent
you from executing code like this:

ObjectA.AssociationToB := ObjectB;

If  this  association  were  to  be  allowed,  and  then  EcoSpace  instance  B  was  deactivated,  ObjectA  would  be  holding  a
reference to an object that no longer has access to its owning EcoSpace. This would result in ObjectA.AssociationToB.Name
:=  'Hello'  attempting  to  write  to  the  cache  of  a  deactivated  EcoSpace,  which  would  most  likely  throw  a
NullReferenceException.

NOTE: At the time of writing this exception is not thrown when calling SetElement, but there are plans to ensure it will be in
future.

5. Change the constructor of fmAuthorArticles again:

constructor Create(AOwner: TComponent; SelectedAuthorID: string);
 
...
 
constructor TfmAuthorArticles.Create(AOwner: TComponent; SelectedAuthorID: string);
var
  SelectedAuthor: Author;
begin
  //Call the base constructor
  inherited Create(AOwner);
 
  //Create a new EcoSpace for this form to use and activate it
  SetEcoSpace(QuickStartVCLEcoSpace.Create);
  EcoSpace.Active := True;
 
  //Get the element using the ID, from this element get an object that
  //may be cast to an Author object
  SelectedAuthor := EcoSpace.ExternalIds.ObjectForId(SelectedAuthorID).AsObject as 
Author;
 
  //Set the element to the author that was passed from the main form
  rhRoot.SetElement(SelectedAuthor);
end;

Instead of  an  Author  instance being  passed to  the  constructor  the  main  form will  now pass  a  string  identifying  the  object.
One of  the ECO services (The External  ID Service) is used to load the specified object  from the data storage, this is then
cast to an Author object.

NOTE:  SetElement  will  accept  the result  of  ObjectForId  directly,  there  is  no need to  convert  it  to  a  business object  (using
.AsObject) and to cast it to an Author.

6. Next the opposite method of ObjectForId will be used in order to retrieve the ID string for the selected author, which will 
then be passed to fmAuthorArticles.

AuthorArticlesForm := TfmAuthorArticles.Create(
  Application,
  EcoSpace.ExternalIds.IdForObject(SelectedAuthor)
  );

7. Now drop a button onto fmAuthorArticles.

8. Name it btnCommit.

9. Set its OnClick event to the following:

EcoSpace.UpdateDatabase;

10. Run the application again.

11. Select an author.

12. Ensure the NumberOfArticles column is visible and make a note of its current value.

13. Click "Show articles".

14. Create a new article and Post the changes.

15. Click the Commit button.

16. Now close the form.

You might notice that this time the NumberOfArticles value has remained unchanged. This clearly shows that the changes
made in the secondary form were completely independent from the values in the main form.

17. Click the Refresh button on the navigator, the NumberOfArticles will now update.

14 Using the DBLookupComboBox 

The next exercise will demonstrate how to set a single-role property on the Article class to an instance of another class using
a TDBLookupComboBox.

1. Activate the Model project.

2. Select the [Model View] tab.

3. Expand the QuickStartPackage node and double-click the class diagram.

4. Add a new class named "ArticleType".

5. Add a UML attribute Name:String with a length of 32.

6. Add a UML attribute ID:AutoInc

7. Now draw an association from ArticleType to the Article class.

8. Expand the property Article.ArticleType

1. Name = ArticleType

2. Multiplicity = 1

9. Expand the property ArticleType.Articles

1. Name = Articles

2. Aggregation = Aggregate (don't allow deletion if any articles are assigned to it)

10. Click the "Update ECO source" button at the top of the model view and then compile.

11. To save us from having to develop a GUI just for this demonstration we will create some ArticleType instances in the 
constructor of the main form. Open Unit1 and in the constructor add the following source code:

constructor TForm1.Create(AOwner);

//Count how many instances of ArticleType exist in the data storage
if EcoSpace.Extents.AllInstances(typeof(ArticleType)).Count = 0 then
begin
  //If there are none then create some and call UpdateDatabase
  ArticleType.Create(EcoSpace).Name := 'VCL.NET';
  ArticleType.Create(EcoSpace).Name := 'ASP.NET';
  ArticleType.Create(EcoSpace).Name := 'WinForms';
  EcoSpace.UpdateDatabase();
end;

12. Switch over to the fmAuthorArticles form.

13. Select ehAuthorArticles.

14. In the Object Inspector click the [...] property editor for the Columns property.

15. Add a new column

1. Name = ArticleType

2. Expression = self.ArticleType (NOTE: "self" should be lowercase)

Now we need a DataSet that will present a list of available article types for the user to choose from.

16. Add a TExpressionHandle to the form.

1. Set its name to "ehArticleTypes".

2. Set its RootHandle to ehRoot.

3. Set its expression to - ArticleType.allInstances->orderBy(name)

4. Set AddDefaultProperties to False.

5. Right-click the component and select "Create default columns".

NOTE: Although the context of  ehArticleType is an instance of the modeled Author class it  is still  possible to switch to the
context  of  the entire  model  by  starting your  expression with  the name of  a  class "ArticleType",  which is  why we must  use
"self.ArticleType" when referring to a member with the same name as a class.

17. Select the ehArticleTypes component.

18. Click the [...] property editor for its Columns property.

19. Add a new column.

1. Name = SelectedArticleType

2. Expression = self (NOTE: Case sensitive)

20. Add a TDataSource to the form.

1. Name = dsArticleTypes

2. DataSet = ehArticleTypes

21. Now add a TDBLookupComboBox to the form

When using a DBLookupComboBox the component will do this

• Show a list of items from the ListSource

• Display the ListField for each item

• (User selects an item)

• Use the KeyField to identify the selected row

• Set the DataSource to point to this row, using the DataField column

I  find  this  quite  easy  to  remember  if  the  properties  in  the  object  inspector  are  sorted  alphabetically.  You  just  start  at  the
bottom of the property list and work your way to the top. Starting with "where do I get the value from?" and ending up with
"where do I put the selected value?".

In a traditional DB application the KeyField would be a primary key on the source table. However, ECO objects are assigned
via their object identity rather than a primary key. Meaning that instead of this

Article.ArticleTypeID := 1234;

you would do this

Article.ArticleType := AnInstanceOfArticleType;

So what should the KeyField be set to? The answer is quite simple. It  needs to be set to a column on ehArticleTypes that
points to the object itself. We need a column with the expression - self.

22. Set the following properties on the combo box.

1. Name = lcbArticleType

2. ListSource = dsArticleTypes

3. ListField = Name

4. KeyField = SelectedArticleType

5. DataSource = dsAuthorArticles

6. DataField = ArticleType

23. Run the application.

24. Select an Author.

25. Click the "Show articles" button.

26. Select an article.

27. Set an article type using the DBLookupComboBox.

28. Select a different article.

29. Select an article type.

30. Click the "Commit" button to save your changes and then close the articles form.

31. Reopen the articles form, note how the values were persisted, and how the value of the DBLookupComboBox changes 
as you scroll through the list of articles.

15 Master detail example 

No  VCL.NET  DB  based  document  would  be  complete  without  the  mandatory  "Master-Detail"  example.  This  example  will
demonstrate  how  to  display  only  articles  for  a  selected  author,  and  also  how  new  articles  added  are  associated  with  the
selected author.

1. Select File->New->Other from the main menu.

2. Select the ECO node in the tree view on the left.

3. Select "New ECO VCL.NET form" and click OK.

4. Name the form fmMasterDetail and then save it as MasterDetailForm.pas

5. In the Project->Options menu ensure that this form is not created automatically by the application.

6. Select the rhRoot component on the new form.

7. In the Object Inspector set its EcoSpaceTypeName to QuickStartVCLEcoSpace

8. Add a TExpressionHandle to the form

1. Name = ehAuthors

2. RootHandle = rhRoot

3. Expression = Author.allInstances

4. AddDefaultProperties = False

9. Edit the columns for ehAuthors and a new column

1. Name = FullName

2. Expression = self.FullName

10. Add a TDataSource to the form

1. Name = dsAuthors

2. DataSet = ehAuthors

11. Add a TDBGrid to the form

1. Name = grdAuthors

2. DataSource = dsAuthors

12. Add a TDBNavigator to the form

1. Name = navAuthors

2. DataSource = dsAuthors

Now to add the detail grid.

13. Add a TExpressionHandle to the form

1. Name = ehAuthorArticles

2. RootHandle = ehAuthors

14. Now click the [...] property editor for the Expression property.

When the OCL editor appears you will notice the following:

1. The context is "Author"

2. UML attributes and associations of the Author class are available.

15. Set the expression to self.Articles and click OK.

16. Now add a TDataSource to the form.

1. Name = dsAuthorArticles

2. DataSet = ehAuthorArticles

17. Now add a TDBGrid and a TDBNavigator to the form and link them both to dsAuthorArticles.

18. Finally, add a button to the main form and add the following source code

...
 
implementation
uses
  AuthorArticlesForm, QuickStartPackageUnit, MasterDetailForm;
 
...
 
procedure TForm1.fmShowMasterDetailExampleClick(Sender: TObject);
var
  Form: TfmMasterDetail;
begin
  Form := TfmMasterDetail.Create(Application, EcoSpace);
  try
    Form.ShowModal;
  finally
    Form.Free;
  end;
end;

Now  run  the  application  and  click  the  Master  Detail  button  on  the  main  form.  You  are  now  able  to  perform  master-detail
operations  by  creating  and  deleting  authors,  and  also  creating  and  deleting  articles.  Note  that  any  new  articles  are
automatically associated with the selected author.

In addition to seeing the full name of the author remember that this is a derived settable member. You can replace the value
by separating the elements of the name with a forward slash.

Mr/John/Smith Jones

16 Sorting data 

The next exercise will outline the steps required to add sorting to the GUI controls displaying the data from an ECO data set.

1. Bring up the form designer for the main form.

2. Select the ehAuthors component in the Object Inspector.

3. Ensure that SupportSorting is True.

4. Now select the DBGrid grdAuthors.

5. In the events list double-click thge OnTitleClick event and enter the following source code:

procedure TForm1.grdAuthorsTitleClick(Column: TColumn);
begin
  //If this is not the same column that was clicked last time then
  //sort alphabetically
  if Column <> SortedColumn then
  begin
    SortedColumn := Column;
    SortDirection := ListSortDirection.Ascending;
  end else
  begin
    //Otherwise sort in the direction opposite of last time
    //the column was clicked
    case SortDirection of
      ListSortDirection.Ascending:
        SortDirection := ListSortDirection.Descending;
 
      ListSortDirection.Descending:
        SortDirection := ListSortDirection.Ascending;
    end;
  end;
 
  ehAuthors.Sort(SortedColumn.FieldName, SortDirection);
end;

6. Finally, in the interface section of the unit add the following two new members to the private section of the form's 
declaration:

  private
    { Private declarations }
    SortedColumn: TColumn;
    SortDirection: ListSortDirection;

Run the application and click some column headers on the grid.

17 Filtering data 

The next exercise will demonstrate how to filter the data presented in the GUI to the user.

1. On the main form select the ehAuthors component in the Object Inspector.

2. Set the FilterMode property to OCL.

3. Now add a standard TEdit control to the form

1. Name it edLastNameFilter

2. Ensure the Text property has an empty value

3. Set its OnChange event implementation to the following source code:

procedure TForm1.edLastNameFilterChange(Sender: TObject);
  function EscapeStringInput(Value: string): string;
  var
    Quote: Char;
  begin
    Quote := '''';
    Result := Value;
    //Backslash is a special character in OCL, so escape it
    Result := StringReplace(Result, '\', '\\', [rfReplaceAll]);
 
    //Don't allow the user to terminate our string, terminate quotes!
    Result := StringReplace(Result, Quote, '\' + Quote, [rfReplaceAll]);
 
    //Put % at the start and end of the string, so that the query
    //denotes "containing"
    Result := '%' + Result + '%';
  end;
 
var
  FilterOCL: string;
begin
  FilterOCL := '';
  if Trim(edLastNameFilter.Text) <> '' then
  begin
    FilterOCL := FilterOCL +
      Format('lastName.sqlLikeCaseInsensitive(''%s'')', 
[EscapeStringInput(edLastNameFilter.Text)]);
  end;
  ehAuthors.Filter := FilterOCL;
end;

When the contents of the TEdit change a new OCL expression is generated and the data set will apply the filter. To achieve
this ECO will  evaluate the given OCL expression against every row in the data set, if  the expression evaluates to True the
row is included in the result set, otherwise it is omitted.

In this example if the word "Smith" is entered into the TEdit then the following OCL will be evaluated against each line.

lastName.sqlLikeCaseInsensitive('%Smith%')

The input provided by the user is modified by the EscapeStringInput function. This is to ensure that

1. The input does not contain malicious characters which could alter the meaning of the OCL (like SQL injection).

2. The value applied to the expression means "contains" rather than "exact match".

To achieve this the following steps are executed:

1. \ is replaced with \\. This is because \ is a special character in OCL used for inserting special characters such as \r\n for 
carriage return + line feed.

2. ' is replaced with \'. This is in case the user enters a name with an apostrophe such as O'Leary. Escaping the single quote 
will inform the OCL that this is a literal character to use rather than terminating the string in the OCL expression.

3. Wrap the value with % and %. These are wildcard characters, just as they are in the SQL "like" statement.

Now run the application, ensuring you have more than one author in the data grid start to type one of their last names into
the edLastNameFilter edit.

18 Validating input 

During the modeling stage a number of  OCL constraints  were defined on the Author  and Article  classes.  If  you recall,  the
return type of these expressions were all Boolean. Evaluating these constraints allows the application to determine whether
or not the object is valid as defined by its constraints.

Next validation will be added to the application. This will prevent any invalid objects from being saved by throwing a custom
EConstraintError.

1. Open the EcoSpaceUnit and view the source.

2. Beneath the definition of QuickStartVCLEcoSpace class define a new subclass of Exception named "EConstraintError";

EConstraintError = class(Exception);

3. Locate the UpdateDatabase method of QuickStartVCLEcoSpace and change the code to read as follows.

procedure QuickStartVCLEcoSpace.UpdateDatabase;
var
  InvalidObjectsExist: Boolean;
  OclExpression: string;
  DirtyObjects: IObjectList;
begin
  //Only continue if the EcoSpace is persistent
  if Assigned(Persistence) and Assigned(DirtyList) then
  begin
    //Get a list of all "Dirty" objects
    DirtyObjects := DirtyList.AllDirtyObjects;
 
    //Using this list evalute an OCL expression that will return all broken
    //constraints
    //  self {a collection of objects, all of which are dirty}
    //  .constraints {every constraint of every object in the list}
    //  ->select {only return constraints that match the following criteria}
    //    (currentConstraint | not currentConstraint)
    //      { "currentConstraint |" defines an alias, "not currentConstraint" selects 
constraints that are false}
    //  ->notEmpty {returns True if the result is not an empty list}
    OclExpression := 'self.constraints->select(currentConstraint | not 
currentConstraint)->notEmpty';
 
    //Evaluate the expression, and get the resulting value as a Boolean
    InvalidObjectsExist := Ocl.Evaluate(DirtyObjects, OclExpression).GetValue<Boolean>;
 
    if InvalidObjectsExist then
      raise EConstraintError.Create('Please correct your errors before saving')
    else
      Persistence.UpdateDatabaseWithList(DirtyObjects);
  end;
end;

4. Locate the Click event code in the main form and change it to read as follows.

procedure TForm1.btnCommitClick(Sender: TObject);
begin
  if ehAuthors.State <> dsBrowse then
    ShowMessage('Please post your changes before committing')
  else
  try
      EcoSpace.UpdateDatabase;
  except
    on Error: EConstraintError do
      ShowMessage(Error.Message);
  end;
end;

5. Run the application.

6. Add a new Author row and enter an item with only a first name and no email address.

7. Post the row.

8. Click the Commit button. You will now see an error message telling you to correct your errors before saving.

At the moment the user is made aware of the fact that they cannot save due to errors, but are unaware of what the errors
are. To rectify this situation a ListBox will be added to the bottom of the form outlining any errors the user needs to correct. A
Delphi  Frame control  will  be  created so  that  this  functionality  may be added to  subsequent  forms with  a  simple  drag/drop
action.

1. File->New->Other

2. Select "Delphi for .NET Projects"

3. Select "New Files"

4. Select "VCL Frame" and click OK.

5. Set the name of the new frame to "EcoErrorsFrame".

6. Add a TLabel to the control:

1. Name = lblErrors

2. Align = alTop

3. Caption = 'Errors'

7. Add a TListBox to the control:

1. Name = lbErrors

2. Align = alClient

This  control  will  show a  list  of  broken constraints  for  all  "Dirty"  objects  in  the  current  EcoSpace.  To do this  it  needs to  be
provided with an instance of an EcoSpace; it then needs to subscribe to be notified whenever the DirtyList changes; it then

needs to subscribe to be notified whenever the list of broken constraints changes for any of these dirty objects.

8. First add "Eco.Subscription" and "Eco.ObjectRepresentation" to the uses clause of the new frame control.

9. Add an "EcoSpace" property.

private
  FEcoSpace: DefaultEcoSpace;
  procedure SetEcoSpace(AEcoSpace: DefaultEcoSpace);
public
  property EcoSpace: DefaultEcoSpace read FEcoSpace write SetEcoSpace;

10. Next add an EventSubscriber to the private section of the control. This is the object that handles notifications from ECO.

private
  Subscriber: EventSubscriber;
public
  constructor Create(AOwner: TComponent); override;
  destructor Destroy; override;
 
...
 
constructor TEcoErrorsFrame.Create(AOwner: TComponent);
begin
  inherited;
  Subscriber := EventSubscriber.Create;
  //Specify which method to execute whenever a subscription changes
  Include(Subscriber.Receive, BrokenConstraintsChanged);
end;
 
destructor TEcoErrorsFrame.Destroy;
begin
  Subscriber.CancelAllSubscriptions;
  inherited;
end;

11. The subscriber will execute a method named BrokenConstraintsChanged any time something it is subscribed to changes.

1. Clear the TListBox.

2. Define an OclExpression that will select all broken constraints for a specific object instance.

3. Loop through each dirty object and get its broken constraints.

4. Get a collection of IConstraintCheck by evaluating the OclExpression.

5. Add the name of the constraint to the TListBox.Items

private
  procedure BrokenConstraintsChanged(Sender: TObject; Args: EventArgs);
 
...
 
procedure TEcoErrorsFrame.BrokenConstraintsChanged(Sender: TObject; Args: EventArgs);
var
  OclExpression: string;
  DirtyObject: IObject;
  BrokenConstraints: IElementCollection;
  ConstraintCheck: IConstraintCheck;
begin

  //#1 Clear the TListBox
  lbErrors.Items.BeginUpdate;
  lbErrors.Clear;
 
  //#2 Define an OclExpression that will select all broken constraints for a specific 
object instance
  OclExpression := 'self.constraints->select(currentConstraint | not currentConstraint)';
 
  //#3 Loop through each dirty object and get its broken constraints
  for DirtyObject in EcoSpace.DirtyList.AllDirtyObjects do
  begin
    //#4 Get a list of broken constraints.  The ".AsCollection" at the end casts to 
IElementCollection for us.
    BrokenConstraints := EcoSpace.Ocl.EvaluateAndSubscribe(DirtyObject, OclExpression, 
Subscriber.AsISubscriber, Subscriber.AsISubscriber).GetAsCollection;
    //#5 Add the name of the constraint to the TListBox.Items
    for ConstraintCheck in BrokenConstraints do
    begin
      lbErrors.Items.Add(ConstraintCheck.Constraint.Name);
    end;
  end;
  lbErrors.Items.EndUpdate;
 
  ReplaceSubscriptions;
end;

12. The above method is called whenever the list of DirtyObjects changes, or a constraint for one of those objects changes 
state between valid/invalid. As the list of dirty objects may have changed our subscriptions will need to be replaced, and 
this is the purpose of the call to "ReplaceSubscriptions" at the end of the method.

1. Cancel previous subscriptions.

2. Subscribe to the dirty list changing.

3. Get a list of all dirty objects.

4. Create an OclExpression to evaluate the constraints on a collection of objects.

5. Subscribe to this expression on the list of dirty objects.

private
  procedure ReplaceSubscriptions;
 
...
 
procedure TEcoErrorsFrame.ReplaceSubscriptions;
var
  OclExpression: string;
  DirtyObjects: IObjectList;
begin
  //Cancel previous subscriptions
  Subscriber.CancelAllSubscriptions;
 
  //Subscribe to the dirty list changing
  EcoSpace.DirtyList.Subscribe(Subscriber.AsISubscriber);
 
  //Get a list of all dirty objects
  DirtyObjects := EcoSpace.DirtyList.AllDirtyObjects;
 
  //Create an OclExpression to evaluate the constraints on a collection of objects
  OclExpression := 'self.constraints->select(currentConstraint | not currentConstraint)';
 
  //Subscribe to this expression on the list of dirty objects
  EcoSpace.Ocl.EvaluateAndSubscribe(DirtyObjects, OclExpression, 
Subscriber.AsISubscriber, Subscriber.AsISubscriber);
end;

13. When the EcoSpace property is set by its owning form the subscriptions need to be placed.

procedure TEcoErrorsFrame.SetEcoSpace(AEcoSpace: DefaultEcoSpace);
begin
  if AEcoSpace = nil then
    raise ArgumentNullException.Create('AEcoSpace');
 
  FEcoSpace := AEcoSpace;
  ReplaceSubscriptions;
end;

14. Save the control as "EcoErrorsFrameUnit.pas".

15. Right-click the control and select "Add to palette".

1. Name = EcoErrorsFrame

2. Palette page = Templates

16. Switch back to the main form.

17. Expand the [Templates] category in the tool palette.

18. Add an EcoErrorsFrame to the bottom of the form.

1. Name = EcoErrorsFrame1

2. Align = alBottom

19. Set the EcoSpace property for EcoErrorsFrame1 when the EcoSpace of the main form is set.

procedure TForm1.SetEcoSpace(AEcoSpace: QuickStartVCLEcoSpace);
begin
  FEcoSpace := AEcoSpace;
  // Note that you need to set the ecospace property for each Referencehandle you add to 
the form
  rhRoot.EcoSpace := FEcoSpace;
  EcoErrorsFrame1.EcoSpace := FEcoSpace;
end;

20. Run the application again.

21. Add a new Author row and Post it.

22. You will now observe a list of errors the user needs to correct.

23. Give the Author a first name and Post again.

24. You will notice the constraint "First name required" will disappear from the error list.

19 Summary 

This article has demonstrated how to use a single ECO class package to form a business model for a VCL.NET application.
Instances of  the contained business classes were then created and modified using TDataSet  descendants as if  they were
rows of data rather than objects. Using this approach it is possible to use the classic "data" approach for writing applications
but with additional benefits such as inheritance, virtual methods, and ECO state machines.

In the next article of this series the XML persistence mapper will be replaced with a mapper that not only modifies data within
an RDBMS database, but also creates and keeps up-to-date the database meta data .

